
19 Beliefs

A POMDP is an MDP with state uncertainty. The agent receives a potentially
imperfect observation of the current state rather than the true state. From the past
sequence of observations and actions, the agent develops an understanding of
the world. This chapter discusses how the belief of the agent can be represented
by a probability distribution over the underlying state. Various algorithms are
presented for updating our belief based on the observation and action taken by
the agent.1 We can perform exact belief updates if the state space is discrete or if 1 Different methods for belief up-

dating are discussed in the context
of robotic applications by S. Thrun,
W. Burgard, and D. Fox, Probabilis-
tic Robotics. MIT Press, 2006.

certain linear Gaussian assumptions are met. In cases where these assumptions
do not hold, we can use approximations based on linearization or sampling.

19.1 Belief Initialization

There are different ways to represent our beliefs. In this chapter, we will discuss
parametric representations where the belief distribution is represented by a set of
parameters of a fixed distribution family, such as the categorical or themultivariate
normal distribution. We will also discuss non-parametric representations where
the belief distribution is represented by particles, or points sampled from the state
space. Associated with different belief representations are different procedures
for updating the belief based on the action taken by the agent and the observation.

Before the agent takes any actions or makes any observations, we start with
an initial belief distribution. If we have some prior information about where the
agent might be in the state space, we can encode this in the initial belief. We
generally want to use diffuse initial beliefs in the absence of information to avoid
being overly confident in the agent being in a region of the state space where it
might not actually be. A strong initial belief focused on states that are far from
the true state can lead to poor state estimates even after many observations.

A diffuse initial prior can cause difficulties, especially for non-parametric
representations of the belief, where the state space can only be very sparsely
sampled. In some cases, it may be useful to wait until an informative observation
is made to initialize our beliefs. For example, in robot navigation problems, we
might want to wait until the sensors detect a known landmark and then initialize
the belief appropriately. The landmark can help narrow down the relevant region
of the state space so we can focus our sampling of the space in the area consistent
with the landmark observation. Example 19.1 illustrates this concept.

Consider an autonomous car equipped with a localization system that uses
camera, radar, and lidar data to track its position. The car is able to identify a
unique landmark at a range r and bearing θ from its current pose.

car

landmark
r

θ

The range and bearing measurements have zero-mean Gaussian noise with
variance νr and νθ , respectively, and the landmark is known to be at (x, y).
Given a measurement r and θ, we can produce a distribution over the car’s
position (x̂, ŷ) and orientation ψ̂:

r̂ ∼ N (r, νr) θ̂ ∼ N (θ, νθ) φ̂ ∼ U (0, 2π)

x̂ ← x + r̂ cos φ̂ ŷ← y + r̂ sin φ̂ ψ̂← φ̂− θ̂ + π

where φ̂ is the angle of the car from the landmark in the global frame.

Example 19.1. Generating an initial
non-parametric belief based on a
landmark observation. In this case,
the autonomous car could be any-
where in a ring around the land-
mark:

landmark

19.2 Discrete State Filter

Figure 19.1. A dynamic decision
network for the POMDP problem
formulation.

In a POMDP, the agent does not directly observe the underlying state of the
environment. Instead, the agent receives an observation, which belongs to some
observation space O, at each time step. The probability of observing o given the
agent took action a and transitioned to state s′ is given by O(o | a, s′). If O

2021-01-07 22:28:48-08:00, revision abe81c1, draft: send comments to bugs@algorithmsbook.com

is continuous, then O(o | a, s′) is a probability density. Figure 19.1 shows the
dynamic decision network associated with POMDPs. Algorithm 19.1 provides an
implementation of the POMDP data structure.

struct POMDP

γ # discount factor

𝒮 # state space

𝒜 # action space

𝒪 # observation space

T # transition function

R # reward function

O # observation function

TRO # sample transition, reward, and observation

end

Algorithm 19.1. A data struc-
ture for POMDPs. We will use
the TRO field to sample the next
state, reward, and observation
given the current state and action:
s′, r, o = TRO(s, a).

A kind of inference known as recursive Bayesian estimation can be used to update
our belief distribution over the current state given the most recent action and
observations. We use b(s) to represent the probability (or probability density for
continuous state spaces) assigned to state s. A particular belief b belongs to a
belief space B, which contains all possible beliefs.

When the state and observation spaces are finite, we can use a discrete state filter
to perform this inference exactly. Beliefs for problems with discrete state spaces
can be represented using categorical distributions, where a probability mass is
assigned to each state. This categorical distribution can be represented as a vector
of length |S| and is often called a belief vector. In cases where b can be treated as
a vector, we will use b. In this case, B ⊂ R|S|. Sometimes B is referred to as a
probability simplex or belief simplex.

Because a belief vector represents a probability distribution, the elements must
be strictly non-negative and must sum to one:

b(s) ≥ 0 for all s ∈ S ∑
s

b(s) = 1 (19.1)

In vector form, we have
b ≥ 0 1>b = 1 (19.2)

The belief space for a POMDPwith three states is given in figure 19.2. An example
discrete POMDP problem is given in example 19.2.

1

1

1

b1

b2

b3

Figure 19.2. The set of valid be-
lief vectors for problems with three
states. Though the state space is
discrete, the belief space is contin-
uous.

If an agent with belief b takes an action a and receives an observation o, the
new belief b′ can be calculated as follows due to the independence assumptions

374 chapter 19. beliefs

The crying baby problem is a simple POMDPwith two states, three actions, and
two observations. Our goal is to care for a baby, and we do so by choosing at
each time step whether to feed the baby, sing to it, or ignore it.

The baby becomes hungry over time. One does not directly observe
whether the baby is hungry, but instead receives a noisy observation in
the form of whether or not the baby is crying. A hungry baby cries 80 % of
the time, whereas a sated baby cries 10 % of the time. Singing to the baby
yields a perfect observation. The state, action, and observation spaces are:

S = {hungry, sated}
A = {feed, sing, ignore}
O = {crying,quiet}

The transition dynamics are:

T(sated | hungry, feed) = 100 %

T(hungry | hungry, sing) = 100 %

T(hungry | hungry, ignore) = 100 %

T(sated | sated, feed) = 100 %

T(hungry | sated, sing) = 10 %

T(hungry | sated, ignore) = 10 %

The reward function assigns −10 reward if the baby is hungry and −5
reward for feeding the baby. Thus, feeding a hungry baby results in −15
reward. Singing to a sated baby makes it happy to receive the attention, and
thus yields 5 reward. Singing to a hungry baby takes extra effort, and thus
incurs −2 reward. As baby caretakers, we seek the optimal infinite horizon
policy with discount factor γ = 0.9.

Example 19.2. The crying baby
problem is a simple POMDP used
to demonstrate decision making
with state uncertainty.

st st+1

ot

rt

atFeed

Reward

Hungry

Crying

19.3. l inear gaussian filter 375

in figure 19.1:

b′(s′) = P(s′ | b, a, o) (19.3)
∝ P(o | b, a, s′)P(s′ | b, a) (19.4)
∝ O(o | a, s′)P(s′ | b, a) (19.5)
∝ O(o | a, s′)∑

s
P(s′ | a, b, s)P(s | b, a) (19.6)

∝ O(o | a, s′)∑
s

T(s′ | s, a)b(s) (19.7)

An example of updating discrete beliefs is given in example 19.3, and the belief
update is implemented in algorithm 19.2. The success of the belief update depends
upon having accurate observation and transition models. In cases where these
models are not knownwell, it is generally advisable to use simplified models with
more diffuse distributions to help prevent overconfidence that leads to brittleness
in the state estimates.

19.3 Linear Gaussian Filter

We can adapt equation (19.7) to handle continuous state spaces as follows:

b′(s′) ∝ O(o | a, s′)
∫

T(s′ | s, a)b(s)ds (19.8)

The integration above can be challenging unless we make some assumptions
about the form of T, O, and b. A special type of filter known as the Kalman filter
(algorithm 19.3)2 provides an exact update under the assumption that T and O 2 Named after the Hungarian-

American electrical engineer
Rudolf E. Kálmán (1930–2016)
who was involved in the early
development of this filter.

are linear–Gaussian and b is Gaussian:3

3 R. E. Kálmán, ‘‘A New Approach
to Linear Filtering and Prediction
Problems,’’ ASME Journal of Ba-
sic Engineering, vol. 82, pp. 35–45,
1960. A comprehensive overview
of the Kalman filter and its vari-
ants is provided by Y. Bar-Shalom,
X. R. Li, and T. Kirubarajan, Estima-
tion with Applications to Tracking and
Navigation. Wiley, 2001.

T(s′ | s, a) = N (s′ | Tss + Taa, Σs) (19.9)
O(o | s′) = N (o | Oss′, Σo) (19.10)

b(s) = N (s | µb, Σb) (19.11)

The Kalman filter begins with a predict step, which uses the transition dynamics
to get a predicted distribution with the following mean and covariance:

µp ← Tsµb + Taa (19.12)

Σp ← TsΣbT>s + Σs (19.13)

376 chapter 19. beliefs

The crying baby problem (example 19.2) assumes a uniform initial belief
state: [b(sated), b(hungry)] = [0.5, 0.5].

Suppose we ignore the baby and the baby cries. We update our belief
according to equation (19.7):

b′(sated) ∝ O(crying | ignore, sated)∑
s

T(sated | s, ignore)b(s)

∝ 0.1(0.0 · 0.5 + 0.9 · 0.5)

∝ 0.045

b′(hungry) ∝ O(crying | ignore, hungry)∑
s

T(hungry | s, ignore)b(s)

∝ 0.8(1.0 · 0.5 + 0.1 · 0.5)

∝ 0.440

After normalizing, our new belief is approximately [0.0928, 0.9072]. A crying
baby is likely to be hungry.

Suppose we then feed the baby and the crying stops. Feeding determinis-
tically caused the baby to be sated, so the new belief is [1, 0].

Finally, we sing to the baby and the baby is quiet. Equation (19.7) is used
again to update the belief, resulting in [0.9759, 0.0241]. A sated baby only
becomes hungry 10 % of the time, and this percentage is further reduced by
not observing any crying.

Example 19.3. Discrete belief up-
dating in the crying baby problem.

function update(b::Vector{Float64}, 𝒫, a, o)

𝒮, T, O = 𝒫.𝒮, 𝒫.T, 𝒫.O

b′ = similar(b)

for (i′, s′) in enumerate(𝒮)

po = O(a, s′, o)

b′[i′] = po * sum(T(s, a, s′) * b[i] for (i, s) in enumerate(𝒮))

end

if sum(b′) ≈ 0.0

fill!(b′, 1)

end

return normalize!(b′, 1)

end

Algorithm 19.2. A method that
updates a discrete belief based on
equation (19.7), where b is a float-
ing point vector. If the given obser-
vation has a zero likelihood, a uni-
form distribution is returned.

In the update step, we use this predicted distribution with the current observa-
tion to update our belief:

K← ΣpO>s
(

OsΣpO>s + Σo

)−1
(19.14)

µb ← µp + K
(

o−Osµp

)
(19.15)

Σb ← (I−KOs)Σp (19.16)

where K is called the Kalman gain.

struct KalmanFilter

μb # mean vector

Σb # covariance matrix

end

function update(b::KalmanFilter, 𝒫, a, o)

μb, Σb = b.μb, b.Σb

Ts, Ta, Os = 𝒫.Ts, 𝒫.Ta, 𝒫.Os

Σs, Σo = 𝒫.Σs, 𝒫.Σo

# predict

μp = Ts*μb + Ta*a

Σp = Ts*Σb*Ts' + Σs

# update

K = Σp*Os'/(Os*Σp*Os' + Σo)

μb′ = μp + K*(o - Os*μp)

Σb′ = (I - K*Os)*Σp

return KalmanFilter(μb′, Σb′)

end

Algorithm 19.3. The Kalman filter,
which updates beliefs in the form
of Gaussian distributions. The cur-
rent belief is represented by μb and
Σb, and the POMDP contains ma-
trices that define linear Gaussian
dynamics.

Kalman filters are often applied to systems that do not actually have linear–
Gaussian dynamics and observations. A variety of different modifications to the
basic Kalman filter have been proposed to better accommodate such systems.4 4 S. Thrun, W. Burgard, and D. Fox,

Probabilistic Robotics. MIT Press,
2006.

19.4 Extended Kalman Filter

The extended Kalman filter (EKF), is a simple extension of the Kalman filter to
problems whose dynamics are nonlinear with Gaussian noise:

T(s′ | s, a) = N (s′ | fT(s, a), Σs) (19.17)
O(o | s′) = N (o | fO(s′), Σo) (19.18)

where fT(s, a) and fO(s′) are differentiable functions.

378 chapter 19. beliefs

Consider the following linear–Gaussian transition and observation functions:

T(s′ | s, a) = N
(

s′ | s + a,
1
10

[
1 1/2

1/2 1

])

O(o | s′) = N
(

o | s′,
1

20

[
1 −1/2
−1/2 2

])

Suppose our initial belief is b = N ([−0.75, 1], I). We take action
a = [0.5,−0.5] and observe o = [0.3, 0.5].

Using the Kalman filter update equations, our new belief is:

b′ = N
([

0.184
0.571

]
,

[
0.037 −0.011
−0.011 0.050]

])

using the Kalman gain:

K =

[
0.789 0.110
0.128 0.716

]

Note that because the observation dynamics are more certain, the new
belief is centered closer to the observation than to the ideal s′ predicted by
the transition dynamics applied to the previous belief mean and action.

Example 19.4. Using a Kalman fil-
ter to update the belief in a linear–
Gaussian POMDP. While in this
problem the state, action, and ob-
servations are all in R2, there is
no requirement that they have the
same dimensionality.

Exact belief updates through nonlinear dynamics are not guaranteed to pro-
duce new Gaussian beliefs, as shown in figure 19.3. The extended Kalman filter
uses a local linear approximation to the nonlinear dynamics, thereby producing
a new Gaussian belief that approximates the true updated belief. We can use
similar update equations as the Kalman filter, but must compute the matrices Ts

and Os at every iteration based on our current belief.
The local linear approximation to the dynamics, or linearization, is given by first-

order Taylor expansions in the form of Jacobians.5 For the state matrix, the Taylor 5 The Jacobian of a multivariate
function f with n inputs and m out-
puts is an m× n matrix where the
(i, j)th entry is ∂ fi/∂xj.

expansion is conducted at µb and the current action, whereas for the observation
matrix, it is computed at the predicted mean µp = fT(µb).

The extended Kalman filter is implemented in algorithm 19.4. Although it is
an approximation, it is fast and performs well on a variety of real-world problems.
The EKF does not generally preserve the true mean and variance of the posterior,
and it does not model multimodal posterior distributions.

struct ExtendedKalmanFilter

μb # mean vector

Σb # covariance matrix

end

import ForwardDiff: jacobian

function update(b::ExtendedKalmanFilter, 𝒫, a, o)

μb, Σb = b.μb, b.Σb

fT, fO = 𝒫.fT, 𝒫.fO

Σs, Σo = 𝒫.Σs, 𝒫.Σo

# predict

μp = fT(μb, a)

Ts = jacobian(s->fT(s, a), μb)

Os = jacobian(fO, μp)

Σp = Ts*Σb*Ts' + Σs

# update

K = Σp*Os'/(Os*Σp*Os' + Σo)

μb′ = μp + K*(o - fO(μp))

Σb′ = (I - K*Os)*Σp

return ExtendedKalmanFilter(μb′, Σb′)

end

Algorithm 19.4. The extended
Kalman filter, an extension of the
Kalman filter to problems with
nonlinear Gaussian dynamics. The
current belief is represented by
mean μb and covariance Σb, and the
POMDP defines the nonlinear dy-
namics via the mean transition dy-
namics fT and mean observation
dynamics fO. The Jacobians are ob-
tained using the ForwardDiff.jl

package.

19.5 Unscented Kalman Filter

The unscented Kalman filter (UKF)6 is another extension to the Kalman filter to

6 S. J. Julier and J. K. Uhlmann, ‘‘Un-
scented Filtering and Nonlinear Es-
timation,’’ Proceedings of the IEEE,
vol. 92, no. 3, pp. 401–422, 2004.

problems that are nonlinear with Gaussian noise.7 Unlike the extended Kalman

7 According to Uhlmann, the term
‘‘unscented’’ comes from a label
on a deodorant container that he
saw on someone’s desk. He used
that term to avoid calling it the
Uhlmann filter. For a historical per-
spective, see ethw.org/First-Han
d:The_Unscented_Transform.

mx+ b

Linear Dynamics Nonlinear Dynamics

f̃ (x)
=

f (µ) +
f ′(µ)(x−

µ)

Linear Approximation

Figure 19.3. Updating a Gaussian
belief with a linear transform (left)
produces another Gaussian distri-
bution. Updating a Gaussian belief
with a nonlinear transform (cen-
ter) does not in general produce
a Gaussian distribution. The ex-
tended Kalman filter uses a lin-
ear approximation of the transform
(right), thereby producing another
Gaussian distribution that approx-
imates the posterior.

filter, the unscented Kalman filter is derivative-free, and relies on a determinis-
tic sampling strategy to approximate the effect of a distribution undergoing a
(typically nonlinear) transformation.

The unscented Kalman filter was developed to estimate the effect of passing a
random variable x through a nonlinear distribution f(x), producing the random
variable x′. We would like to estimate the mean µ′ and covariance Σ′ of the
distribution over x′. The unscented transform allows for more information of p(x)
to be used than the mean µ and covariance Σ of the distribution over x.8 8 We need not necessarily assume

that the prior distribution is Gaus-
sian.

An unscented transform passes a set of sigma points S through f and uses the
transformed points to approximate the transformed mean µ′ and covariance Σ′.
The mean and covariance are reconstructed using a vector of weights w:

µ′ = ∑
i

wisi (19.19)

Σ′ = ∑
i

wi(si − µ′)(si − µ′)> (19.20)

These weights must sum to 1 in order to provide an unbiased estimate, but need
not all be positive.

The updated mean and covariance matrix given by the unscented transform
through f are thus:

µ′ = ∑
i

wif(si) (19.21)

Σ′ = ∑
i

wi
(
f(si)− µ′

)(
f(si)− µ′

)> (19.22)

An example set of sigma points and associated weight vector is given in exam-
ple 19.5.

A simple set of 2n sigma points for an n-dimensional distribution is given by:

s2i = µ+
√

nΣi for i in 1 : n

s2i−1 = µ−
√

nΣi for i in 1 : n

where the i subscript indicates the ith column of the square root matrix.
If we use the weights wi = 1/2n, the reconstructed mean is:

∑
i

wisi =
n

∑
i=1

1
2n

(
µ+
√

nΣi

)
+

1
2n

(
µ−
√

nΣi

)
=

n

∑
i=1

1
n
µ = µ

and the reconstructed covariance is:

∑
i

wi(si − µ′)(si − µ′)> = 2
n

∑
i=1

1
2n

(√
nΣi

)(√
nΣi

)>
=

1
n

n

∑
i=1

(√
nΣi

)(√
nΣi

)>
=
√

Σ
√

Σ
>

= Σ

Example 19.5. A simple set of
sigma points and weights that can
be used to reconstruct the mean
and covariance of a distribution.

The square root of a matrix A is
a matrix B such that BB> = A. In
Julia, the sqrt method produces a
matrix C such that CC = A, which
is not the same. One common
square root matrix can be obtained
from the Cholesky decomposition.

Sigma points andweights can be selected to incorporate additional information
about the prior distribution. Suppose we always have a set of sigma points and
weights that produce a mean µ′ and covariance Σ′. We can construct a new set of
sigma points by additionally including the mean µ′ as a sigma point. We scale
the sigma points such that the covariance is preserved. This results in a scaled set

382 chapter 19. beliefs

of sigma points with different higher-order moments, but the same mean and
covariance. Example 19.6 derives a set of sigma points by applying this technique
to the sigma points in example 19.5.

We can include the mean µ in the sigma points from example 19.5 to obtain
a new set of 2n + 1 sigma points:

s1 = µ

s2i = µ+

(√
n

1− w1
Σ

)
i
for i in 1 : n

s2i+1 = µ−
(√

n
1− w1

Σ

)
i
for i in 1 : n

where w1 is the weight on the first sigma point. The weights on the remaining
sigma points are (1− w1)/2n. The reconstructed mean is still µ, and the
reconstructed covariance is still Σ.

We can vary w1 to produce different sets of sigma points. Setting w1 > 0
causes the sigma points to spread away from the mean. Setting w1 < 0 moves
the sigma points closer to the mean.

Example 19.6. An example of how
a set of sigma points can be pa-
rameterized, allowing the designer
to capture additional information
about the prior distribution.

A common set of sigma points include the mean µ ∈ Rn and an additional 2n
points formed from perturbations of µ in directions determined by the covariance
matrix Σ:

s1 = µ (19.23)

s2i = µ+

(√
(n + λ)Σ

)
i
for i in 1 : n (19.24)

s2i+1 = µ−
(√

(n + λ)Σ

)
i
for i in 1 : n (19.25)

These sigma points are associated with the weights:

wi =

 λ
n+λ for i = 1

1
2(n+λ)

otherwise
(19.26)

The scalar spread parameter λ determines how far the sigma points are spread
from themean. It is recommended to use λ = 2, which is optimal for matching the
fourth moment of Gaussian distributions. Several sigma point sets for different
values of λ are shown in figure 19.4.

−4 −2 0 2 4
−4

−2

0

2

4

x1

x 2

λ = −0.5

−4 −2 0 2 4
x1

λ = 0.0

−4 −2 0 2 4
x1

λ = 2.0

−4 −2 0 2 4
x1

λ = 4.0 Figure 19.4. The effect of varying
λ on the sigma points from equa-
tion (19.23) generated for a Gaus-
sian distribution with zero mean
and covariance
Σ = [1 1/2; 1/2 2].

The unscented Kalman filter performs two unscented transformations, one for
the prediction step and one for the observation update. The original Kalman filter
computes the Kalman gain using the observation covariance Os. The unscented
Kalman filter does not have this observation matrix, and instead calculates the
Kalman gain using a cross covariance matrix9 using both prediction values and 9 A covariancematrix measures the

variance between components of
the same multi-dimensional vari-
able. A cross covariance matrix
measures the variance between
two multi-dimensional variables.
In algorithm 19.5, we compute a
cross covariance matrix Σpo.

observation values. Algorithm 19.5 provides an implementation.

19.6 Particle Filter

Discrete problems with large state spaces or continuous problems with dynamics
that are not well approximated by the linear–Gaussian assumption of the Kalman
filter must often resort to approximation techniques to represent the belief and to
perform the belief update. One common approach is to use a particle filter, which
represents the belief state as a collection of states.10 Each state in the approximate 10 A tutorial on particle filters is

provided by M. S. Arulampalam, S.
Maskell, N. Gordon, and T. Clapp,
‘‘A Tutorial on Particle Filters for
Online Nonlinear / Non-Gaussian
Bayesian Tracking,’’ IEEE Transac-
tions on Signal Processing, vol. 50,
no. 2, pp. 174–188, 2002.

belief is called a particle.
A particle filter is initialized by selecting or randomly sampling a collection

of particles that represent the initial belief. The belief update for a particle filter
with m particles begins by propagating each state si by sampling from the tran-
sition distribution to obtain a new state s′i with probability T(s′i | si, a). The new
belief is constructed by drawing m particles from the propagated states weighted
according to the observation function wi = O(o | a, s′). This procedure is given
in algorithm 19.6. Example 19.7 illustrates an application of a particle filter.

struct UnscentedKalmanFilter

μb # mean vector

Σb # covariance matrix

λ # spread parameter

end

function unscented_transform(μ, Σ, f, λ, ws)

n = length(μ)

Δ = sqrt((n + λ) * Σ)

S = [μ]

for i in 1:n

push!(S, μ + Δ[:,i])

push!(S, μ - Δ[:,i])

end

S′ = f.(S)

μ′ = sum(w*s for (w,s) in zip(ws, S′))

Σ′ = sum(w*(s - μ′)*(s - μ′)' for (w,s) in zip(ws, S′))

return (μ′, Σ′, S, S′)

end

function update(b::UnscentedKalmanFilter, 𝒫, a, o)

μb, Σb, λ = b.μb, b.Σb, b.λ

fT, fO = 𝒫.fT, 𝒫.fO

n = length(μb)

ws = [λ / (n + λ); fill(1/(2*(n + λ)), 2n)]

# predict

μp, Σp, Sp, Sp′ = unscented_transform(μb, Σb, s->fT(s,a), λ, ws)

Σp += 𝒫.Σs

# update

μo, Σo, So, So′ = unscented_transform(μp, Σp, fO, λ, ws)

Σo += 𝒫.Σo

Σpo = sum(w*(s - μp)*(s′ - μo)' for (w,s,s′) in zip(ws, So, So′))

K = Σpo / Σo

μb′ = μp + K*(o - μo)

Σb′ = Σp - K*Σo*K'

return UnscentedKalmanFilter(μb′, Σb′, λ)

end

Algorithm 19.5. The unscented
Kalman filter, an extension of the
Kalman filter to problems with
nonlinear Gaussian dynamics. The
current belief is represented by
mean μb and covariance Σb. The
POMDP parameterizes the nonlin-
ear dynamics with the mean tran-
sition dynamics fT and mean ob-
servation dynamics fO. The sigma
points used in the unscented trans-
forms are controlled by the scalars
α, β, and κ.

Suppose we want to determine our position based on imperfect distance
measurements to radio beacons whose locations are known. We remain
approximately still for a few steps to collect independent measurements. The
particle filter states are our potential locations. We can compare the ranges
we would expect to measure for each particle to the observed ranges.

We assume individual range observations from each beacon are observed
with zero-mean Gaussian noise. Our particle transition function adds zero-
mean Gaussian noise since we remain only approximately still.

The images below show the evolution of the particle filter. The rows cor-
respond to different numbers of beacons. The red dot indicates our true
location and the blue dots are particles. The circles indicate the positions
consistent with noiseless distance measurements from each sensor.

y

t = 1 t = 2 t = 3 t = 4

y

x

y

x x x

Three beacons are required to accurately identify our location. A strength
of the particle filter is that it is able to represent the multimodal distributions
that are especially apparent when there are only one or two beacons.

Example 19.7. A particle filter ap-
plied to different beacon configura-
tions.

struct ParticleFilter

states # vector of state samples

end

function update(b::ParticleFilter, 𝒫, a, o)

T, O = 𝒫.T, 𝒫.O

states = [rand(T(s, a)) for s in b.states]

weights = [O(a, s′, o) for s′ in states]

D = SetCategorical(states, weights)

return ParticleFilter(rand(D, length(states)))

end

Algorithm 19.6. A belief updater
for particle filters, which updates
a vector of states representing the
belief based on the agent’s action a

and its observation o.

In problems with discrete observations, we can also perform particle belief
updates with rejection. Here, each state si in the filter is propagated to obtain
successor states s′i as before. However, observations oi are simultaneously sampled
for each successor state. Any oi that does not equal the true observation o is
rejected, and a new s′i and oi are sampled until the observations match. This
particle filter with rejection is implemented in algorithm 19.7.

struct RejectionParticleFilter

states # vector of state samples

end

function update(b::RejectionParticleFilter, 𝒫, a, o)

T, O = 𝒫.T, 𝒫.O

states = similar(b.states)

i = 1

while i ≤ length(states)

s = rand(b.states)

s′ = rand(T(s,a))

if rand(O(a,s′)) == o

states[i] = s′

i += 1

end

end

return RejectionParticleFilter(states)

end

Algorithm 19.7. Updating a par-
ticle filter with rejection, which
forces sampled states to match the
input observation o.

As the number of particles in a particle filter increases, the distribution repre-
sented by the particles approaches the true posterior distribution. Unfortunately,
particle filters can fail in practice. Low particle coverage and the stochastic nature
of the resampling procedure can cause there to be no particles near the true

state. This problem of particle deprivation can be somewhat mitigated by several
strategies. A motivational example is given in example 19.8.

Spelunker Joe is lost in a grid-based maze. He lost his lantern, so can only
observe his surroundings by touch. At any givenmoment, Joe can tell whether
his location in the maze has walls in each cardinal direction. Joe is fairly
confident in his ability to feel walls, so he assumes his observations are
perfect.

Joe uses a particle filter to track his belief over time. At some point he stops
to rest. He continues to run his particle filter to update his belief.

The initial belief has one particle in each grid location that matches his
current observation of a wall to the north and south. Spelunker Joe does not
move and does not gain new information, so his belief should not change
over time. Due to the stochastic nature of resampling, subsequent beliefs may
not contain all of the initial states. Over time, his belief will continue to lose
states until it only contains a single state. It is possible that this state is not
where Spelunker Joe is located.

Example 19.8. A particle filter
run for enough time can lose par-
ticles in relevant regions of the
state space due to the stochastic na-
ture of resampling. The problem
is more pronounced with fewer
particles or when the particles are
spread over a large state space.

19.7 Particle Injection

Particle injection involves injecting random particles to protect against particle
deprivation. Algorithm 19.8 injects a fixed number of particles from a broader
distribution, such as a uniform distribution over the state space.11 While particle

11 For robotic localization problems,
it is common practice to inject
particles from a uniform distribu-
tion over all possible robot poses,
weighted by the current observa-
tion.

injection can help prevent particle deprivation, it also reduces the accuracy of the
posterior belief represented by the particle filter.

Instead of using a fixed number of injected particles at each update, we can
take a more adaptive approach. When the particles are all being given very low
weight, then we generally want to inject more particles. It might be tempting
to choose the number of injected particles based solely on the mean weight of

struct InjectionParticleFilter

states # vector of state samples

m_inject # number of samples to inject

D_inject # injection distribution

end

function update(b::InjectionParticleFilter, 𝒫, a, o)

T, O, m_inject, D_inject = 𝒫.T, 𝒫.O, b.m_inject, b.D_inject

states = [rand(T(s, a)) for s in b.states]

weights = [O(a, s′, o) for s′ in states]

D = SetCategorical(states, weights)

m = length(states)

states = vcat(rand(D, m - m_inject), rand(D_inject, m_inject))

return InjectionParticleFilter(states, m_inject, D_inject)

end

Algorithm 19.8. Particle filter
update with injection, in which
m_inject particles are sampled
from the injection distribution
D_inject to reduce the risk of par-
ticle deprivation.

the current set of particles. However, doing so can make the success of the filter
sensitive to naturally low observation probabilities in the early periods when the
filter is still converging or in moments of high sensor noise.12 12 S. Thrun,W. Burgard, and D. Fox,

Probabilistic Robotics. MIT Press,
2006.

Algorithm 19.9 presents an adaptive injection algorithm that keeps track of two
exponential moving averages of the mean particle weight and bases the number
of injections on their ratio.13 If wmean is the current mean particle weight, the two 13 D. E. Goldberg and J. Richardson,

‘‘An Experimental Comparison of
Localization Methods,’’ in Interna-
tional Conference on Genetic Algo-
rithms, 1987.

moving averages are updated according to

wfast ← wfast + α(wmean − wfast) (19.27)
wslow ← wslow + α(wmean − wslow) (19.28)

where 0 ≤ αslow < αfast ≤ 1.
The number of injected samples in a given iteration is obtained by comparing

the fast and slow mean particle weights:14 14 Note that bxe denotes the integer
nearest to x.

minject =

⌊
m max

(
0, 1− ν

wfast
wslow

)⌉
(19.29)

The scalar ν ≥ 1 allows us to tune the injection rate.

19.8 Summary

• Partially observable Markov decision processes (POMDPs) extend MDPs to
include state uncertainty.

mutable struct AdaptiveInjectionParticleFilter

states # vector of state samples

w_slow # slow moving average

w_fast # fast moving average

α_slow # slow moving average parameter

α_fast # fast moving average parameter

ν # injection parameter

D_inject # injection distribution

end

function update(b::AdaptiveInjectionParticleFilter, 𝒫, a, o)

T, O = 𝒫.T, 𝒫.O

w_slow, w_fast, α_slow, α_fast, ν, D_inject =

b.w_slow, b.w_fast, b.α_slow, b.α_fast, b.ν, b.D_inject

states = [rand(T(s, a)) for s in b.states]

weights = [O(a, s′, o) for s′ in states]

w_mean = mean(weights)

w_slow += α_slow*(w_mean - w_slow)

w_fast += α_fast*(w_mean - w_fast)

m = length(states)

m_inject = round(Int, m * max(0, 1.0 - ν*w_fast / w_slow))

D = SetCategorical(states, weights)

states = vcat(rand(D, m - m_inject), rand(D_inject, m_inject))

b.w_slow, b.w_fast = w_slow, w_fast

return AdaptiveInjectionParticleFilter(states,

w_slow, w_fast, α_slow, α_fast, ν, D_inject)

end

Algorithm 19.9. A particle fil-
ter with adaptive injection, which
maintains fast and slow expo-
nential moving averages w_fast

and w_slow of the mean parti-
cle weight with smoothness fac-
tors α_fast and α_slow, respec-
tively. Particles are only injected
if the fast moving average of the
mean particle weight is less than
1/ν of the slow moving average.
Recommended values from the
original paper are α_fast = 0.1,
α_slow = 0.001, and ν = 2.

Spelunker Joe from example 19.9 now moves one tile east and moves all
particles in his particle filter one tile to the east as well. He now only senses
walls to the north and east, and unfortunately, this observation does not agree
with any of the updated particles in his filter. He decides to use adaptive
injection to fix his particle deprivation problem. Below we see how his filter
injects particles from a uniform random distribution, along with the values
for the fast and slow filters.

wslow = 1.0
w f ast = 1.0

wslow = 0.99
w f ast = 0.7

wslow = 0.98
w f ast = 0.49

wslow = 0.97
w f ast = 0.34

wslow = 0.96
w f ast = 0.24

wslow = 0.95
w f ast = 0.17

wslow = 0.94
w f ast = 0.12

wslow = 0.93
w f ast = 0.1

Example 19.9. A particle filter with
adaptive injection α_slow = 0.01,
α_fast = 0.3, and ν = 2.0 start-
ing from a deprived state with 16
identical particles. The moving av-
erages are initialized to 1 to reflect
a long period of observations that
perfectly match every particle in
the filter. Over the next iterations,
these moving averages change at
different rates based on the quan-
tity of particles that match the ob-
servation.

• The uncertainty requires agents in a POMDP to maintain a belief over their
state.

• Beliefs for POMDPs with discrete state spaces can be represented using cate-
gorical distributions and can be updated analytically.

• Beliefs for linear Gaussian POMDPs can be represented using Gaussian distri-
butions and can also be updated analytically.

• Beliefs for nonlinear continuous POMDPs can also be represented using Gaus-
sian distributions but cannot typically be updated analytically. In this case the
extended Kalman filter and the unscented Kalman filter can be used.

• Continuous problems can sometimes be modeled assuming they are linear
Gaussian.

• Particle filters approximate the belief with a large collection of state particles.

19.9 Exercises

Exercise 19.1. Can every MDP be framed as a POMDP?

Solution: Yes. The POMDP formulation extends the MDP formulation by introducing state
uncertainty in the form of the observation distribution. Any MDP can be framed as a
POMDP with O = S and O(o | a, s′) = (o = s′).

Exercise 19.2. An autonomous vehicle represents its belief over its position using a mul-
tivariate normal distribution. It comes to a rest at a traffic light, and the belief updater
continues to run while it sits. Over time, the belief concentrates and becomes extremely
confident in a particular location. Why might this be a problem? How might this extreme
confidence be avoided?

Solution: Yes, overconfidence in a belief can be a problemwhen themodels or belief updates
do not perfectly represent reality. The overconfident belief may have converged on a state
that does not match the true state. Once the vehicle moves again, new observations may
be inconsistent with the belief and result in poor estimates. To help address this issue, we
can require that the values of the diagonal elements of the covariance matrix be above
some threshold.

Exercise 19.3. Consider tracking our belief over the dud rate for widgets produced at a
factory. We use a Poisson distribution to model the probability that k duds are produced
in one day of factory operation given that the factory has a dud rate of λ:

P(k | λ) =
1
k!

λke−λ . . .

Suppose that our initial belief over the dud rate follows a gamma distribution:

p(λ | α, β) =
βα

Γ(α)
λα−1e−βλ

where λ ∈ (0, ∞) and the belief is parameterized by the shape α > 0 and the rate β > 0.
After a day of factory operation, we observe that d ≥ 0 duds were produced. Show that
our updated belief over the dud rate is also a gamma distribution.15 15 The gamma distribution is a con-

jugate prior to the Poisson distribu-
tion. A conjugate prior is a family
of probability distributions that re-
main within the same family when
updated with an observation. Con-
jugate priors are useful for model-
ing beliefs because their form re-
mains constant.

Solution: We seek the posterior distribution p(λ | d, α, β), which we can obtain through
Bayes’ rule:

p(λ | d, α, β) ∝ p(d | λ)p(λ | α, β)

∝
1
d!

λde−λ βα

Γ(α)
λα−1e−βλ

∝ λα+d−1e−(β+1)λ

This is a gamma distribution:

p(λ | α + d, β + 1) =
(β + 1)α+d

Γ(α + d)
λα+d−1e−(β+1)λ

∝ λα+d−1e−(β+1)λ

Exercise 19.4. Why are particle filters with rejection not used for updating beliefs in
POMDPs with continuous observations?

Solution: Rejection sampling requires repeatedly sampling the transition and observation
functions until the sampled observation matches the true observation. The probability
of sampling any particular value in a continuous probability distribution is zero, making
rejection sampling run forever. In practice, we would use a finite representation for contin-
uous values such as 64-bit floating point numbers, but rejection sampling can run for an
extremely long time for each particle.

Exercise 19.5. Explain why Spelunker Joe would not benefit from switching to a particle
filter with adaptive injection with ν ≥ 1 in example 19.8.

Solution: The particle filter with adaptive injection injects new particles when νwfast/wslow
is less than 1. Spelunker Joe assumes perfect observations and has a belief with particles
that match his current observation. Thus, every particle has a weight of 1, and so both
wfast and wslow are 1. It follows that wfast/wslow is always 1, leading to no new particles.

Exercise 19.6. Why is the injection rate scalar ν in a particle filter with adaptive injection
typically not set to a value less than 1?

Solution: Particle injection was designed to inject particles when the current observations
have lower likelihood than a historic trend over the observation likelihood. Thus, injection
typically only occurs when the short-term estimate of the mean particle weight wfast is less
than the long-term estimate of the mean particle weight wslow. If ν < 1, then particles can
still be generated even if wfast ≥ wslow, despite indicating that current observations have
higher likelihood than the past average.

Exercise 19.7. Supposewe are performing in-flightmonitoring of an aircraft. The aircraft is
either in a state of normal operation s0 or a state of malfunction s1. We receive observations
through the absence of a warning w0 or the presence of a warning w1. We can choose to
allow the plane to continue to fly m0 or send the plane in for maintenance m1. We have
the following transition and observation dynamics, where we assume the warnings are
independent of the actions given the status of the plane:

T(s0 | s0, m0) = 0.95 O(w0 | s0) = 0.99

T(s0 | s0, m1) = 1 O(w1 | s1) = 0.7

T(s1 | s1, m0) = 1

T(s0 | s1, m1) = 0.98

Given the initial belief b = [0.95, 0.05], compute the updated belief b′ given that we allow
the plane to continue to fly and we observe a warning.

Solution: Using equation (19.7), we update the belief for s0:

b′(s0) ∝ O(w1 | s0)∑
s

T(s0 | s, m0)b(s)

b′(s0) ∝ O(w1 | s0)(T(s0 | s0, m0)b(s0) + T(s0 | s1, m0)b(s1)

b′(s0) ∝ (1− 0.99)(0.95× 0.95 + (1− 1)× 0.05) = 0.009025

We repeat the update for s1:

b′(s1) ∝ O(w1 | s1)∑
s

T(s1 | s, m0)b(s)

b′(s1) ∝ O(w1 | s1)(T(s1 | s0, m0)b(s0) + T(s1 | s1, m0)b(s1))

b′(s1) ∝ 0.7((1− 0.95)× 0.95 + 1× 0.05) = 0.06825

After normalization, we obtain the updated belief:

b′(s0) =
b′(s0)

b′(s0) + b′(s1)
≈ 0.117

b′(s1) =
b′(s1)

b′(s0) + b′(s1)
≈ 0.883

b′ ≈ [0.117, 0.883]

Exercise 19.8. Consider a robot moving along a line with position x, velocity v, and
acceleration a. At each timestep, we directly control the acceleration and we observe the
velocity. The equations of motion for the robot are

x′ = x + v∆t + 1
2 a∆t2

v′ = v + a∆t

Suppose we would like to implement a Kalman filter to update our belief of the robot’s
true state. The state vector is s = [x, v]. Determine Ts, Ta, and Os.

Solution: The transition and observation dynamics can be written in linear form as follows[
x′

v′

]
=

[
1 ∆t
0 1

] [
x
v

]
+

[
1
2 ∆t2

∆t

]
a

o =
[
0 1

] [x
v

]

Through these equations, we can identify Ts, Ta, and Os

Ts =

[
1 ∆t
0 1

]
Ta =

[
1
2 ∆t2

∆t

]
Os =

[
0 1

]
Exercise 19.9. Compute the set of sigma points and weights with λ = 2 for a multivariate
Gaussian distribution with

µ =

[
1
2

]
Σ =

[
4 0
0 2.25

]

Solution: Since we have a two-dimensional Gaussian distribution and we are given λ = 2,
we need to compute 2n + 1 = 5 sigma points. We need to compute the square-root matrix
B =

√
(n + λ)Σ, such that BB> = (n+λ)Σ. Since the scaled covariancematrix is diagonal,

the square-root matrix is simply the element-wise square root of (n + λ)Σ

√
(n + λ)Σ =

√√√√(2 + 2)

[
4 0
0 2.25

]
=

[
4 0
0 3

]

Now, we can compute the sigma points and weights:

s1 =

[
1
2

]
w1 =

2
2 + 2

=
1
2

s2 =

[
1
2

]
+

[
4
0

]
=

[
5
2

]
w2 =

1
2(2 + 2)

=
1
8

s3 =

[
1
2

]
−
[

4
0

]
=

[
−3
2

]
w3 =

1
2(2 + 2)

=
1
8

s4 =

[
1
2

]
+

[
0
3

]
=

[
1
5

]
w4 =

1
2(2 + 2)

=
1
8

s5 =

[
1
2

]
−
[

0
3

]
=

[
1
−1

]
w5 =

1
2(2 + 2)

=
1
8

Exercise 19.10. Using the sigma points and weights from the previous exercise, compute
the updated mean and covariance given by the unscented transform through f(x) =

[2x1, x1x2].

Solution: The transformed sigma points are

f(s1) =

[
2
2

]
f(s2) =

[
10
10

]
f(s3) =

[
−6
−6

]
f(s4) =

[
2
5

]
f(s5) =

[
2
−1

]

We can reconstruct the mean as the weighted sum of transformed sigma points

µ′ = ∑
i

wif(si)

µ′ =
1
2

[
2
2

]
+

1
8

[
10
10

]
+

1
8

[
−6
−6

]
+

1
8

[
2
5

]
+

1
8

[
2
−1

]
=

[
2
2

]

The covariancematrix can be reconstructed from theweighted sum of pointwise covariance
matrices

Σ′ = ∑
i

wi
(
f(si)− µ′

) (
f(si)− µ′

)>
Σ′ =

1
2

[
0 0
0 0

]
+

1
8

[
64 64
64 64

]
+

1
8

[
64 64
64 64

]
+

1
8

[
0 0
0 9

]
+

1
8

[
0 0
0 9

]
=

[
16 16
16 18.25

]

